Exploring the HC2000 ROM circuitry, and adding 64K of flashram.                 

This is the journal of my experiments, and definitely not a DIY guide!  

Parts of this story were already made available.




 Roelof Koning,  2006.
=================================================================================

[image: image1.wmf]The HC2000 has been marketed in (at least) two versions:

One with a 27C256 EPROM on the calculator part of the Printed Circuit Board (PCB, left hand side) together with a 27C128 EPROM on the IF1 interface part. (right hand side). This version also has two RAM chips type 2114 on the interface part of the PCB.
Another version holds only a 27C512 EPROM on the calculator part of the PCB, and one 6116 RAM chip on the disk interface part of the PCB. It appears that the PCB is prepared for both versions, although the '64K' version probably needs some modifications as on my board there are a few extra wires. So the second version has 16K of ROM and 1K of RAM more than the first version!                a retro freak 
My HC2000 has the EPROM type 27C512 on board which holds 64K of data: 



0     - 16383 = the IF1 system


16384 - 32767 = almost identical copy of  0-16 (not used!)


32768 - 49151 = the Spectrum BASIC with some extra code at 14446


49152 - 65535 = the 8K CPM BIOS and bootstrap loader

16K of this ROM is visible for the hardware when BASIC is running, another block of 16K is present when the IF1 is doing its jobs. A third 16K block of this EPROM holds the code used by CPM. In fact this third block holds two times the 8K of CPM code. The fourth 16K block holds an older version of the IF1 code, which in fact is not used at all!

The switching between the lower and upper 32K is done by a RST8 instruction,
which acts on the A15 adress line. The switching between the lower and upper

half of such 32K block is done by bit 0 of port 126, which acts on A14.

The 2K RAM chip type 6116 is addressed as a block of 1K, and found on addr. 10240-11263, 11264-12287, 14336-15359 and 15360-16383, as soon as the lower 32K of the EPROM is paged in. Indeed, the same 1K of RAM is found 4 times in both the IF1 bank or the Copy-IF1 bank.

The conclusion must be that a lot of ROM space is lost due to the design. Not only a whole 16K bank is not used, but also 8K inside the CPM bank is redundant and in the IF1 bank the space from 11263 –16383 could be made available, of which 11263-12287 would be RAM.  

I wanted to make this all available to me on my HC2000, as it is my hobby to program on ‘system-level’.

Phase 1
The reason for the 1K 'barrier' of the 6116 RAM is very simple, the signal line used to address the second half (1-2K), connection A10, is simply connected to 'ground'! Which means that when this 1-2K block is selected, the (low) 0-1K block appears again. This problem is solved by cutting a small copper trace between pins 19 and 20 of the 6116 chip, and wiring the ‘freed’ pin 19 to the Z80’s A10 address line. 
Phase 2

 Here is shown

 how I did the

 phase 2 job.

The ‘mirroring’ of the 6116 chip in the IF1 space comes from the fact that 
the address line A12 is not taken into account by the 74LS138 type decoder,

while A11 and A13 are. The problem is solved by adding this A12 signal by 

means of a few extra logic gates. The picture above shows how a 74LS02 is soldered on top of the 74LS138. Note that also A14 is brought in to make it possible to discriminate between the 0-16K bank and the copy at 16-32K.

[image: image3.wmf]This prevents the RAM from paging in while the fourth EPROM bank is active.
Phase 3

An EPROM chip is electrically selected with two signals, one to the /CE pin (Chip Enable), and one to the /OE pin (Output Enable. For unknown reasons the HC2000 has the Z80’s /MREQ signal connected to /OE, and has the Z80’s /RD signal left out of the decoding. This means that the EPROM presents its data not only during a ‘read’, but also when something is written to it. This rather unusual design prevents the owner to exchange the EPROM with a

FlashRAM, exactly the thing I had in mind. Fortunately my EPROM was already mounted in a socket, and I did not have to arrange that first.

Shown here is half of the phase 3 job.

A 74LS32 chip is soldered ‘piggy back’

on top of another. Only one of the four

gates is used to add /MREQ to the ROM-

select signal.

For a number of reasons I wanted to keep the EPROM also. The main reason is that I need a ROM with a valid content, to make the paging by RST8 work!

So I soldered a 28 pin socket on top of a 32 pin 28F010 FlashRAM chip, leaving pin 22 (/CE) out by bending it sideways. 

I also bent away to a horizontal position, the pins 1 and 2, 31 and 32, and the pins 22 (/OE) and 24 (/CE) of the 28F010 . 

This chip was then mounted in the EPROM socket leaving pins 1, 2, 31 and 32 outside. On top the EPROM is inserted. See the picture for details.

In fact the drawing depicts the situation before I had to replaced the Z80, and mounted it on a socket. Since then the whole ROM contraption is put on another socket just to gain height.

Note that the EPROM goes into the socket on top and has none of its legs bent! Unlike the FlashRAM.....

This shows the construction of the EPROM/FlashRAN
combination. Hote that both /OE's pins are comnected to
/RD. Flash-pin 32 is connected to EPRON-pin28 for +3V
and pin 1 through a diode.




In the world of FlashRAM there exist two different groups, the models that need 12 Volt to be programmed, the 28-series, and the models that work with 5 Volt only, the 29 series.  I used a 28-type, because these can be write-protected by switching the 12V off. The smallest size is 128KB, of which I only use 64 KB. Whether the lower or the upper bank is used, depends on the voltage on pin 2. As the so called ‘boot block’ appeared to be of no use for me, I choose to have the lower bank and wired pin 2 to GROUND. 

To have the Vpp pin 1 connected to ‘something’ on the moments when there is no 12V, I connected it to +5V by means of a diode. As ‘GROUND’ is available on pin 2, I instead could have connected Vpp (pin 1) with a resistor to this GROUND. 

The picture shows the inserted ‘EPROM-tower’. Visible are two black wires connected to /RD and /WR, the two resistors that ‘pull up’ both /CE connections, and the two green wires going to the ROM/Flash switch. The very thin wire goes to the new /CE. 

The 12V (yellow wire) is connected through a push button that must be pushed during programming. This might be replaced by an other switch.

I always work with parts that I find in the junkbox, and had no specific reason to choose the 28F001. There are several other types available in the 28-series and 29-series that can do the job. 

For programming the FlashRAM, I start the HC2000 with the EPROM selected. All works under standard BASIC now, and the progam and data files can be loaded from disk. My program beeps (for about a second) two times, to signal the moments on which the FlashRAM must be switched in and out. So the whole

is done from standard BASIC.
Phase 4

Programming and erasing a FlashRAM 28F001.

Other types of flashram may need other routines!!

;? * TORNADO *     Roelof Koning 2006                           

;------------------------------------------------               

;               For HC2000                                      

; Reading all banks from both EPROM and FlashRAM                

;     Writing all banks into FlashRAM                           

;                 28F001                                       

;0-16 = IF11, 16-32 = IF12, 32-48 = BAS, 48-64 = CPM            

;------------------------------------------------               

           ORG  48000                                           

           DUMP 48000                                           

           JP   readbas           ;0  USR 48000                           

           JP   readcpm           ;1  USR 48003                 

           JP   readif11          ;2  USR .....                 

           JP   readif12          ;3  ... ..etc.                          

           JP   writebas          ;4                            

           JP   writecpm          ;5                            

           JP   writeif11         ;6                            

           JP   writeif12         ;7                            

           JP   identify          ;8                            

           JP   erase             ;9                            

;                             ---- POKE THESE! ----             

loadreg    LD   DE,0              ;start + 31/32                

           LD   HL,49152          ;start + 34/35                

           LD   BC,16384          ;start + 37/38                

           RET                                                  

;?--------------------------------------------------------      

erase      DI                                                   

           CALL beep              ;time to switch ROM!!         

           LD   A,#50             ;reset status                 

           LD   (000),A                                         

           LD   A,#20             ;erase1                       

           LD   (000),A                                         

           LD   A,#D0             ;erase2                       

           LD   (000),A                                         

ready1     LD   A,(000)           ;read status !!               

           BIT  7,A                                             

           JR   Z,ready1          ;1=ready                      

           AND  %00111000         ;keep bits 3,4,5              

           LD   B,0                                             

           LD   C,A               ;now C holds status           

eexit      LD   A,#50             ;reset status                 

           LD   (000),A                                         

           LD   A,#FF             ;back to normal read          

           LD   (000),A                                         

           CALL ready             ;delay + beep                 

           EI                                                   

           RET                                                  

;?--------------------------------------------                  

identify   DI                                                   

           CALL beep                                            

           LD   A,#50             ;reset status                 

           LD   (000),A                                         

           LD   A,#90             ;identify                     

           LD   (000),A                                         

           LD   BC,(0000)         ;read identifs                

           JR   eexit             ;back to nornal               

;?---------------------------------------------                 

readbas    CALL readbas1         ;without EI!                   

           EI                                                   

           RET                                                  

readbas1   DI                                                   

           CALL beep              ;time to switch ROMs!         

           CALL loadreg           ;poked from basic             

           LDIR                                                 

ready      PUSH BC                ;save result                  

           LD   BC,65535          ;max delay                    

delay      DJNZ delay                                           

           DEC  C                                               

           JR   NZ,delay                                        

           CALL beep              ;time to switch again!!       

           POP  BC                                              

           RET                    ;back to basic                

;?----------------------------------------------                

readcpm    CALL readcpm1          ;without EI!                  

           EI                                                   

           RET                                                 

readcpm1   DI                                                   

           CALL beep              ;time to switch ROMs!         

           CALL loadreg           ;poked from basic             

           LD   A,1               ;CPM or IF1-2 bank            

           OUT  (126),A                                         

           LDIR                   ;fetch code                   

           XOR  A                 ;BASIC or IF1_1 bank          

           OUT  (126),A                                         

           JR   ready                                           

;?---------------------------------------------                 

readif11   LD   HL,readbas1       ;same routine                 

           JR   rprep                                           

readif12   LD   HL,readcpm1                                     

rprep      LD   (23789),HL                                      

           RST  8                                               

           DEFB #32               ;make sure IF1 is in          

           EI                     ;     and run (23789)         

           RET                    ;page out here!               

;?=============================================                 

writebas   CALL writebas1                                       

           EI                                                   

           RET                                                  

writebas1  DI                                                   

           CALL beep              ;time to switch ROMs!         

           CALL loadreg                                         

           EX   DE,HL                                           

           CALL dowrit            ;BASIC                        

           JP   ready                                           

;?---------------------------------------------                 

writecpm   CALL writecpm1                                       

           EI                                                   

           RET                                                  

writecpm1  DI                                                   

           CALL beep              ;time to switch ROMs!         

           CALL loadreg                                         

           EX   DE,HL                                           

           LD   A,1                                             

           OUT  (126),A                                         

           CALL dowrit            ;the actual writing           

           XOR  A                                               

           OUT  (126),A                                         

           JR   ready                                           

;?---------------------------------------------                 

writeif11  LD   HL,writebas1      ;same routine!                

           JR   wprep                                           

writeif12  LD   HL,writecpm1      ;also same routine!           

wprep      LD   (23789),HL                                      

           RST  8                                               

           DEFB #32               ;make sure IF1 is in          

           EI                     ;IF1 ROM paged out now!       

           RET                    ;exit MC here                 

;?-----------------------------------------------               

dowrit     LD   A,#50             ;reset status                 

           LD   (DE),A                                          

dowrit1    LD   A,#40             ;write setup                  

           LD   (DE),A                                          

           LD   A,(HL)                                          

           LD   (DE),A                                          

getstat    LD   A,(DE)            ;read status                  

           BIT  7,A                                             

           JR   NZ,getstat        ;until ready                  

           AND  %00011000         ;keep bits 3 and 4            

           JR   NZ,wexit          ;error                        

           INC  HL                                              

           INC  DE                                              

           DEC  BC                                              

           LD   A,B               ;test for last                

           OR   C                                               

           JR   NZ,dowrit                                       

wexit      LD   B,0                                             

           LD   C,A               ;0 or status                  

           LD   A,#50             ;reset status                 

           LD   (000),A                                         

           LD   A,#FF             ;back to normal read          

           LD   (000),A                                         

           RET                                                  

;?--------------------------------------------                  

beep       LD   E,0                                             

           LD   A,16                                            

           LD   HL,1000                                         

beep1      DEC  E                                               

           JR   NZ,beep1                                        

           XOR  16                                              

           OUT  (254),A                                         

beep2      DEC  E                                               

           JR   NZ,beep2                                        

           DEC  HL                                              

           LD   D,A                                             

           LD   A,H                                             

           OR   L                                               

           LD   A,D                                             

           JR   NZ,beep1                                        

           RET                                                  

;the end --------                                               

This shows the construction of the EPROM/FlashRAN
combination. Hote that both /OE's pins are comnected to
/RD. Flash-pin 32 is connected to EPRON-pin28 for +3V
and pin 1 through a diode.








